
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 15.11 which is part of the ‘IMS_34229_IWD_13wk49’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk49’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
15.11
Test Group:
IMS Supplementary Services
ATS Version:
iwd-EUTRA-B2012-03_D13wk49
System Simulator used:
Rohde & Schwarz Multi-RAT Protocol Tester
UE used:
Intel XMM7160
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk49’ release.
Change 1 – IMS_PTC_ImsInfo_Type
	Type name
	IMS_PTC_ImsInfo_Type

	Reason for change
	The value of the Cseq header is incremented for each request within a dialog. As the value of Cseq is to be checked in f_IMS_A_2_1_Invite_CommonChecks and as the current value is accessible only in different places (e.g. when a PRACK is received before the 15.11 flow actually starts) it seems appropriate to define a global variable holding the latest value of Cseq.

	Summary of change
	Added a field called lastSeenValueOfCseq to a record collecting global state information. It is up for discussion if this is the right place to store the Cseq counter and if it is appropriately named in this proposal. 

	TTCN module
	Common/IMS/IMS_Component

	MCC160 Comment
	


Before change

type record IMS_PTC_ImsInfo_Type {                    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    IP_AddrInfo_Type            UE_Address      optional,

    IP_AddrInfo_Type            NW_Address      optional,

    InternetProtocol_Type       TransportProtocol,              /* transport protocol to be used for requests sent in DL

                                                                 * @sic R5s120530 change 1.7 sic@

                                                                 * NOTE: in contrast to R5s120530 and acc. to rejection of R5-130765 at RAN WG5 Meeting #58 in Malta UDP is used for all requests in DL;

                                                                 *       nevertheless "TransportProtocol" is kept in case there are other decisions in the future */

    IPCAN_RAN_Type              RanType         optional,

    integer                     UE_Release      optional,       /* @sic R5s130739 MCC160 implementation: new field to store the UE's release sic@ */

    boolean                     SigCompStarted,

    IMS_UriInfo_Type            UriInfo,                        /* URIs and public user ids */

    charstring                  HomeDomainName,

    charstring                  PrivateUserId,

    SipUrl                      ContactUrl      optional,       /* contact address as provided in contact header of the REGISTER message;

                                                                   to be used e.g. for Request-URI of INVITE for MT call */

    integer                     BranchIndex,                    /* internal index used to generate unique branches */

    integer                     CallIdCnt,                      /* counter to achieve unique callids e.g. for SMS MESSAGE sent to the UE @sic R5-133705 sic@ */

    charstring                  Pcscf,                          /* P-CSCF domain name or IP address corresponding to current IMS-PTC */

    IMS_PTC_CseqInfo_Type       Cseq,

    IMS_PTC_RegistrationInfo_Type       Registration,

    IMS_PTC_DialogInfo_Type             Dialog,

    IMS_SecurityScheme_Type     SecurityScheme                  /* IMS security or GIBA acc. to 34.229-2 Table A.6a; depends on PICS setting and may depend on test case */

  };
After change

  type record IMS_PTC_ImsInfo_Type {                    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    IP_AddrInfo_Type            UE_Address      optional,

    IP_AddrInfo_Type            NW_Address      optional,

    InternetProtocol_Type       TransportProtocol,              /* transport protocol to be used for requests sent in DL

                                                                 * @sic R5s120530 change 1.7 sic@

                                                                 * NOTE: in contrast to R5s120530 and acc. to rejection of R5-130765 at RAN WG5 Meeting #58 in Malta UDP is used for all requests in DL;

                                                                 *       nevertheless "TransportProtocol" is kept in case there are other decisions in the future */

    IPCAN_RAN_Type              RanType         optional,

    integer                     UE_Release      optional,       /* @sic R5s130739 MCC160 implementation: new field to store the UE's release sic@ */

    boolean                     SigCompStarted,

    IMS_UriInfo_Type            UriInfo,                        /* URIs and public user ids */

    charstring                  HomeDomainName,

    charstring                  PrivateUserId,

    SipUrl                      ContactUrl      optional,       /* contact address as provided in contact header of the REGISTER message;

                                                                   to be used e.g. for Request-URI of INVITE for MT call */

    integer                     BranchIndex,                    /* internal index used to generate unique branches */

    integer                     CallIdCnt,                      /* counter to achieve unique callids e.g. for SMS MESSAGE sent to the UE @sic R5-133705 sic@ */

    charstring                  Pcscf,                          /* P-CSCF domain name or IP address corresponding to current IMS-PTC */

    integer                     lastSeenValueOfCseq,   

    IMS_PTC_CseqInfo_Type       Cseq,

    IMS_PTC_RegistrationInfo_Type       Registration,

    IMS_PTC_DialogInfo_Type             Dialog,

    IMS_SecurityScheme_Type     SecurityScheme                  /* IMS security or GIBA acc. to 34.229-2 Table A.6a; depends on PICS setting and may depend on test case */

  };
Change 2 – cs_IMS_ImsInfo_Init
	Template name
	cs_IMS_ImsInfo_Init

	Reason for change
	To not leave the global variable defined in Change 1 uninitialized 

	Summary of change
	Added a statement initializing with zero to template cs_IMS_ImsInfo_Init which is called in function f_IMS_PTC_Init

	TTCN module
	Common/IMS/IMS_Component

	MCC160 Comment
	


Before change

  template (value) IMS_PTC_ImsInfo_Type cs_IMS_ImsInfo_Init(template (omit) IPCAN_RAN_Type p_RanType,

                                                            template (omit) integer p_UE_Release,

                                                            charstring p_Pcscf,

                                                            charstring p_HomeDomainName,

                                                            charstring p_PrivateUserId,

                                                            IMS_SecurityScheme_Type p_SecurityScheme) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    UE_Address := omit,

    NW_Address := omit,

    TransportProtocol := udp,

    RanType := p_RanType,

    UE_Release := p_UE_Release,

    SigCompStarted := false,

    UriInfo := cs_IMS_UriInfo_Init,

    HomeDomainName := px_IMS_HomeDomainName,

    PrivateUserId := px_IMS_Private_UserId,

    ContactUrl := omit,

    BranchIndex := 0,

    CallIdCnt := 0,

    Pcscf := p_Pcscf,

    Cseq := {

      Register := 0,

      DialogRemote := 0,

      DialogLocal := 0

    },

    Registration := {

      CallId := omit

    },

    Dialog := {

      ToTag := omit

    },

    SecurityScheme := p_SecurityScheme

  };
After change

  template (value) IMS_PTC_ImsInfo_Type cs_IMS_ImsInfo_Init(template (omit) IPCAN_RAN_Type p_RanType,

                                                            template (omit) integer p_UE_Release,

                                                            charstring p_Pcscf,

                                                            charstring p_HomeDomainName,

                                                            charstring p_PrivateUserId,

                                                            IMS_SecurityScheme_Type p_SecurityScheme) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT) */

    UE_Address := omit,

    NW_Address := omit,

    TransportProtocol := udp,

    RanType := p_RanType,

    UE_Release := p_UE_Release,

    SigCompStarted := false,

    UriInfo := cs_IMS_UriInfo_Init,

    HomeDomainName := px_IMS_HomeDomainName,

    PrivateUserId := px_IMS_Private_UserId,

    ContactUrl := omit,

    BranchIndex := 0,

    CallIdCnt := 0,

    Pcscf := p_Pcscf,

    lastSeenValueOfCseq := 0,  

    Cseq := {

      Register := 0,

      DialogRemote := 0,

      DialogLocal := 0

    },

    Registration := {

      CallId := omit

    },

    Dialog := {

      ToTag := omit

    },

    SecurityScheme := p_SecurityScheme

  };
Change 3 – f_IMS_MOCallHoldResume_AnnexC8
	Function name
	f_IMS_MOCallHoldResume_AnnexC8

	Reason for change
	1. The Route header for a re-INVITE request is different from the Route header of an INVITE request creating a dialog in that the Route set of a dialog is taken from the Record-Route header of a previously received response, and the Route header of a new request is then populated from the Route set. 
2. The called function f_IMS_A_2_1_Invite_CommonChecks  needs to be fed additional parameters such that it can properly check the incoming re-INVITE
3. The value of the Cseq header is incremented for each request within a dialog. As the value of Cseq is to be checked in f_IMS_A_2_1_Invite_CommonChecks and as the current value is accessible only in different places (e.g. when a PRACK is received before the 15.11 flow actually starts) it seems appropriate to define a global variable holding the latest value of Cseq. The actual definition of this new global variable is shown in Change 1. In this function, we need to maintain it.
4. Function f_IMS_MOCallHoldResume_AnnexC8 is called twice, once for putting the call on hold, and once for resuming it again. The two calls to it differ in the first parameter indicating Hold or Resume, respectively. This information needs to be forwarded to embedded functions fl_IMS_MOCallHoldResume_AnnexC8_CheckBandwidth and fl_IMS_MOCallHoldResume_AnnexC8_CheckSendRecvMode as these functions show different behavior for Hold vs Resume.  

	Summary of change
	1. Populated the expected Route header with the four entries the SS put into the Record-Route header of preceding 183 Session Progress response, taken in reverse order
Note: a corresponding change for the UPDATE request seems in order
2. Added parameters for Condition A5 (re-INVITE) and the data structures describing the original INVITE request and its 183 Session progress response. Change 4 shows what these three additional parameters are used for.
3. Updated the value of the global variable mirroring the latest value of Cseq by taking it from the latest message exchanged, assuming this was an INVITE request.
4. Replaced fixed indication for Hold by variable p_HoldResume indicating the actual scenario (Hold or Resume).

	TTCN module
	Common/IMS/IMS_Procedures_CallControl

	MCC160 Comment
	


Before change

function f_IMS_MOCallHoldResume_AnnexC8(IMS_CallHoldResume_Type p_HoldResume,

                                          INVITE_Request p_InviteRequest,

                                          MessageHeader p_MessageHeader_Prev183Response,

                                          SDP_Message p_SDP_OriginalOffer,

                                          template (omit) integer p_PrevSessionVersion := omit) runs on IMS_PTC return SDP_Message

  {

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) MessageHeader v_MessageHeader_Update;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) SDP_Message v_SDP_HoldResumeResponse;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var INVITE_Request v_InviteRequest;

    var UPDATE_Request v_UpdateRequest;

    var MessageHeader v_MessageHeader;

    var SDP_Message v_SDP_HoldResumeOffer;

    var integer v_PrevSessionVersion;

    var SipUrl v_RequestSipUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_Prev183Response);           // get the contact URI out of the Contact header being sent to the UE in the earlier 183 response

    var SipUrl v_ToHeaderSipUrl := f_Addr_Union_GetSipUrl(p_InviteRequest.msgHeader.toField.addressField);     // SIP URI in the To header of the INVITE needs to be the same SIP URI as been used in the previous INVITE

    // Steps 1 and 2 - Receive Invite Request or Update Request

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ToHeaderSipUrl);

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderRX(p_InviteRequest, p_MessageHeader_Prev183Response);

    alt {

      // INVITE

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_RequestSipUrl, v_MessageHeader_Invite, cr_MessageBody_SDP)))

        -> value v_IMS_DATA_REQ

        {

          v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

          v_MessageHeader := v_InviteRequest.msgHeader;

          v_SDP_HoldResumeOffer := v_InviteRequest.messageBody.sdpMessageBody;

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ);
        }

      // UPDATE

      [] IMS_Server.receive(car_IMS_Update_Request(cr_UPDATE_Request(v_RequestSipUrl, v_MessageHeader_Update, cr_MessageBody_SDP(?))))

        -> value v_IMS_DATA_REQ

        {

          v_UpdateRequest := v_IMS_DATA_REQ.Request.Update;

          v_MessageHeader := v_UpdateRequest.msgHeader;

          v_SDP_HoldResumeOffer := v_UpdateRequest.messageBody.sdpMessageBody;

        }

    }

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Annex C.8 step 1");

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // Check version number:

    if (isvalue(p_PrevSessionVersion)) {

      v_PrevSessionVersion := valueof(p_PrevSessionVersion);

    } else {

      v_PrevSessionVersion := str2int(p_SDP_OriginalOffer.origin.session_version);

    }

    f_IMS_SDP_CheckSessionVersion(v_SDP_HoldResumeOffer, v_PrevSessionVersion);

    // Check bandwidth

    fl_IMS_MOCallHoldResume_AnnexC8_CheckBandwidth(HOLD, v_SDP_HoldResumeOffer);

    // Check send/receive mode

    fl_IMS_MOCallHoldResume_AnnexC8_CheckSendRecvMode(HOLD, v_SDP_HoldResumeOffer, p_SDP_OriginalOffer);

    if (isbound(v_InviteRequest)) {

      // Step 2. Send 100 Trying

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    }

    // Step 3. Respond with 200 OK (A.3.1 condition A5: Any response sent by the UE within a dialog

    v_SDP_HoldResumeResponse := f_IMS_BuildSDP_AnnexC8_Step3(p_HoldResume, v_SDP_HoldResumeOffer);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                   f_IMS_OtherResponse_200_MessageHeaderTX(v_MessageHeader, cs_ContentTypSDP),

                                                                   cs_MessageBody_SDP(v_SDP_HoldResumeResponse))));

    if (isbound(v_InviteRequest)) {

      // Step 4. Receive ACK

      v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

      IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    }

    return v_SDP_HoldResumeOffer;

  }
After change

  function f_IMS_MOCallHoldResume_AnnexC8(IMS_CallHoldResume_Type p_HoldResume,

                                          INVITE_Request p_InviteRequest,

                                          MessageHeader p_MessageHeader_Prev183Response,

                                          SDP_Message p_SDP_OriginalOffer,

                                          template (omit) integer p_PrevSessionVersion := omit) runs on IMS_PTC return SDP_Message

  {

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) MessageHeader v_MessageHeader_Update;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) SDP_Message v_SDP_HoldResumeResponse;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var INVITE_Request v_InviteRequest;

    var UPDATE_Request v_UpdateRequest;

    var MessageHeader v_MessageHeader;

    var SDP_Message v_SDP_HoldResumeOffer;

    var integer v_PrevSessionVersion;

    var SipUrl v_RequestSipUrl := f_MessageHeader_GetContactSipUrl(p_MessageHeader_Prev183Response);           // get the contact URI out of the Contact header being sent to the UE in the earlier 183 response

    var SipUrl v_ToHeaderSipUrl := f_Addr_Union_GetSipUrl(p_InviteRequest.msgHeader.toField.addressField);     // SIP URI in the To header of the INVITE needs to be the same SIP URI as been used in the previous INVITE

    // Steps 1 and 2 - Receive Invite Request or Update Request

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ToHeaderSipUrl);

        var RouteBody_List v_RouteBodyList := {p_MessageHeader_Prev183Response.recordRoute.routeBody[3],p_MessageHeader_Prev183Response.recordRoute.routeBody[2],p_MessageHeader_Prev183Response.recordRoute.routeBody[1],p_MessageHeader_Prev183Response.recordRoute.routeBody[0]};
    v_MessageHeader_Invite.route.routeBody := v_RouteBodyList;     

    v_MessageHeader_Update := f_IMS_UpdateRequest_MessageHeaderRX(p_InviteRequest, p_MessageHeader_Prev183Response);

    alt {

      // INVITE

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_RequestSipUrl, v_MessageHeader_Invite, cr_MessageBody_SDP)))

        -> value v_IMS_DATA_REQ

        {

          v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

          v_MessageHeader := v_InviteRequest.msgHeader;

          v_SDP_HoldResumeOffer := v_InviteRequest.messageBody.sdpMessageBody;

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, A_2_1_A5, -, -, -, -, -, p_InviteRequest, p_MessageHeader_Prev183Response); 

        }

      // UPDATE

      [] IMS_Server.receive(car_IMS_Update_Request(cr_UPDATE_Request(v_RequestSipUrl, v_MessageHeader_Update, cr_MessageBody_SDP(?))))

        -> value v_IMS_DATA_REQ

        {

          v_UpdateRequest := v_IMS_DATA_REQ.Request.Update;

          v_MessageHeader := v_UpdateRequest.msgHeader;

          v_SDP_HoldResumeOffer := v_UpdateRequest.messageBody.sdpMessageBody;

        }

    }

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Annex C.8 step 1");

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    vc_IMS_Global.ImsInfo.lastSeenValueOfCseq := v_MessageHeader.cSeq.seqNumber;  

    // Check version number:

    if (isvalue(p_PrevSessionVersion)) {

      v_PrevSessionVersion := valueof(p_PrevSessionVersion);

    } else {

      v_PrevSessionVersion := str2int(p_SDP_OriginalOffer.origin.session_version);

    }

    f_IMS_SDP_CheckSessionVersion(v_SDP_HoldResumeOffer, v_PrevSessionVersion);

    // Check bandwidth

    fl_IMS_MOCallHoldResume_AnnexC8_CheckBandwidth(p_HoldResume, v_SDP_HoldResumeOffer);                          

    // Check send/receive mode

    fl_IMS_MOCallHoldResume_AnnexC8_CheckSendRecvMode(p_HoldResume, v_SDP_HoldResumeOffer, p_SDP_OriginalOffer); 

    if (isbound(v_InviteRequest)) {

      // Step 2. Send 100 Trying

      IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    }

    // Step 3. Respond with 200 OK (A.3.1 condition A5: Any response sent by the UE within a dialog

    v_SDP_HoldResumeResponse := f_IMS_BuildSDP_AnnexC8_Step3(p_HoldResume, v_SDP_HoldResumeOffer);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200,

                                                                   f_IMS_OtherResponse_200_MessageHeaderTX(v_MessageHeader, cs_ContentTypSDP),

                                                                   cs_MessageBody_SDP(v_SDP_HoldResumeResponse))));

    if (isbound(v_InviteRequest)) {

      // Step 4. Receive ACK

      v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

      IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    }

    return v_SDP_HoldResumeOffer;

  }
Change 4 – f_IMS_A_2_1_Invite_CommonChecks
	Function name
	f_IMS_A_2_1_Invite_CommonChecks

	Reason for change
	This function runs into fatal error cases when properly feeding condition A_2_1_A5 into it (see Change 1) because the listed checks are not implemented yet. 

	Summary of change
	Implemented the five missing checks (implementing check #3 and #4 in one comparison). As these checks need to access the original INVITE request and its 183 Session Progress response, those are fed into this function. 
Note the embedded comment on the third new check: the proposed check is very strict, and we might want to be more flexible.

	TTCN module
	Common/IMS/IMS_Procedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                           boolean p_A3 := false,

                                           boolean p_A8 := false,

                                           boolean p_A10 := false,

                                           boolean p_A11 := false,

                                           boolean p_A12 := false) runs on IMS_PTC

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    var MessageHeader v_MessageHeader := p_IMS_INVITE_REQ.Request.Invite.msgHeader;

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_AcceptContactParams := {};

    var InternetProtocol_Type v_Protocol := p_IMS_INVITE_REQ.RoutingInfo.Protocol;

    var SipUrl v_SipUrlFrom := f_Addr_Union_GetSipUrl(v_MessageHeader.fromField.addressField);

    var SipUrl v_SipUrlPreferred;

    var template SipUrl v_ExpectedFromUrl := omit;

    var PPreferredIDValue_List v_PreferredIDValueList;

    var CharStringList_Type v_PreferredPublicUserIdList;

    var boolean v_AllowFQDN := (p_Context != A_2_1_A6);  // for A6 there is no FQDN for the UE

    var boolean v_CheckRport := (p_Context == A_2_1_A6);

    var integer i;

    var charstring v_IMEI;

    var template (present) charstring v_SipInstanceValue;

    if (ispresent(v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams)) {

      v_ContactParams := v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams;

    }

    if (ispresent(v_MessageHeader.acceptContact.acValues[0].acRcParams)) {

      v_AcceptContactParams := v_MessageHeader.acceptContact.acValues[0].acRcParams;

    }

    // ****  Common checks ****

    f_IMS_MessageHeader_CheckVia(v_MessageHeader, v_Protocol, v_AllowFQDN, v_CheckRport);   // check whether via contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);                     // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

    if (p_Context != A_2_1_A5) {

      f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);      // check that there is no To tag

    }

    // ****  Check URIs in From header, P-Preferred-Identity header and further headers ****

    select (p_Context) {

      case (A_2_1_A4) {

        // Check Call-Id

        if (match(v_MessageHeader.callId.callid, f_IMS_PTC_ImsInfo_GetCallIdRegistration())) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'CallId' in INVITE (has to be different than in previous REGISTER)");

        }

        // Additional checks regarding SIP URI in the From header @sic R5-130750 change 3 sic@

        if (not f_IMS_PublicUserId_Match(f_IMS_PTC_ImsInfo_GetAllIMPUs(), v_SipUrlFrom)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid 'From'");

        }

        if (ispresent(v_MessageHeader.pPreferredID.pPreferredIDValueList)) {

          v_PreferredIDValueList := v_MessageHeader.pPreferredID.pPreferredIDValueList;

          for (i:=0; i < lengthof(v_PreferredIDValueList); i := i+1) {

            v_PreferredPublicUserIdList[i] := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_PreferredIDValueList[i]));

          }

          if (not f_IMS_PublicUserId_Match(v_PreferredPublicUserIdList, v_SipUrlFrom)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'From': does not match with P-Preferred-Identity");

          }

        }

      }

      case (A_2_1_A5) {

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of Call-ID not implemented yet (value of Call-ID as in any previous request in the same dialog)");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of CSeq not implemented yet ()");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From addr-spec shall be checked");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From tag shall be checked");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Request-URI shall be checked");
      }

      case (A_2_1_A6) {

        v_IMEI := hex2str(px_IMEI_Def); // @sic R5s130756 change 11 sic@

        v_SipInstanceValue := """" & "<urn:gsma:imei:" & substr(v_IMEI, 0, 8) & "-" & substr(v_IMEI, 8, 6) & "-" & substr(v_IMEI, 14, 1) & ">" & """";

        if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("+sip.instance", v_SipInstanceValue))) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "IMEI code of the UE, coded as specified in draft-montemurro-gsma-imei-urn, not matching");

        }

      }

      case (A_2_1_A7) {

        v_ExpectedFromUrl := f_IMS_PTC_ImsInfo_GetSipUriRX(emergencyId, *);

        if (not match(v_SipUrlFrom, v_ExpectedFromUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'From'");

        }

        v_SipUrlPreferred := f_Addr_Union_GetSipUrl(v_MessageHeader.pPreferredID.pPreferredIDValueList[0]);

        if (not match(v_SipUrlPreferred, v_ExpectedFromUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'P-Preferred-Identity'");

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    // ****  Check further Contact feature-param acc. to further conditions of A.2.1:  ****

    //

    if (p_A3) {                         // A.1.2 condition A3: MTSI

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted)) or        // @sic R5s130738: string needs to be quoted sic@

          not match(v_AcceptContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted))) {  // @sic R5s130738: string needs to be quoted sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param");

      }

    }

    if (p_A8) {                         // A.1.2 condition A8: check geolocation and geolocationRouting @sic R5-133151 sic@

      /* NOTE: v_MessageHeader.geolocation and geolocationRouting are checked already by template returned by f_IMS_InviteRequest_MessageHeaderRX:

       *       geolocation and geolocationRouting to be present. @sic R5-133151 sic@ */

//      FatalError(__FILE__, __LINE__, "A.2.1 condition A8: geolocation needs to be checked regarding correct setting of locationURI (not implemented yet)");

      if (not match(v_MessageHeader.geolocationRouting.state, GEOLOCATION_ROUTING_YES_E)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A8: Wrong geolocation-routing");

      }

    }

    if (p_A10) {                        // A.1.2 condition A10: UE indicates video media feature tag in REGISTER and INVITE request

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10: Missing feature-param");

      }

    }

    if (p_A11) {                        // A.1.2 condition A11: INVITE for creating a video call and UE supports video media feature tag

      if (not match(v_AcceptContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A11: Missing feature-param");

      }

    }

    if (p_A12) {                        // A.1.2 condition A12: INVITE for creating a voice call and UE supports g.3gpp.srvcc-alerting media feature tag

      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("g.3gpp.srvcc-alerting", omit))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A12: Missing feature-param");

      }

    }

  }
After change

  function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                           boolean p_A3 := false,

                                           boolean p_A8 := false,

                                           boolean p_A10 := false,

                                           boolean p_A11 := false,

                                           boolean p_A12 := false,

                                           template INVITE_Request p_InviteRequest := *,

                                           template MessageHeader p_183_header := *) runs on IMS_PTC  

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    var MessageHeader v_MessageHeader := p_IMS_INVITE_REQ.Request.Invite.msgHeader;

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_AcceptContactParams := {};

    var InternetProtocol_Type v_Protocol := p_IMS_INVITE_REQ.RoutingInfo.Protocol;

    var SipUrl v_SipUrlFrom := f_Addr_Union_GetSipUrl(v_MessageHeader.fromField.addressField);

    var SipUrl v_SipUrlPreferred;

    var template SipUrl v_ExpectedFromUrl := omit;

    var PPreferredIDValue_List v_PreferredIDValueList;

    var CharStringList_Type v_PreferredPublicUserIdList;

    var boolean v_AllowFQDN := (p_Context != A_2_1_A6);  // for A6 there is no FQDN for the UE

    var boolean v_CheckRport := (p_Context == A_2_1_A6);

    var integer i;

    var charstring v_IMEI;

    var template (present) charstring v_SipInstanceValue;

    if (ispresent(v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams)) {

      v_ContactParams := v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams;

    }

    if (ispresent(v_MessageHeader.acceptContact.acValues[0].acRcParams)) {

      v_AcceptContactParams := v_MessageHeader.acceptContact.acValues[0].acRcParams;

    }

    // ****  Common checks ****

    f_IMS_MessageHeader_CheckVia(v_MessageHeader, v_Protocol, v_AllowFQDN, v_CheckRport);   // check whether via contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);                     // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

    if (p_Context != A_2_1_A5) {

      f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);      // check that there is no To tag

    }

    // ****  Check URIs in From header, P-Preferred-Identity header and further headers ****

    select (p_Context) {

      case (A_2_1_A4) {

        // Check Call-Id

        if (match(v_MessageHeader.callId.callid, f_IMS_PTC_ImsInfo_GetCallIdRegistration())) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'CallId' in INVITE (has to be different than in previous REGISTER)");

        }

        // Additional checks regarding SIP URI in the From header @sic R5-130750 change 3 sic@

        if (not f_IMS_PublicUserId_Match(f_IMS_PTC_ImsInfo_GetAllIMPUs(), v_SipUrlFrom)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid 'From'");

        }

        if (ispresent(v_MessageHeader.pPreferredID.pPreferredIDValueList)) {

          v_PreferredIDValueList := v_MessageHeader.pPreferredID.pPreferredIDValueList;

          for (i:=0; i < lengthof(v_PreferredIDValueList); i := i+1) {

            v_PreferredPublicUserIdList[i] := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_PreferredIDValueList[i]));

          }

          if (not f_IMS_PublicUserId_Match(v_PreferredPublicUserIdList, v_SipUrlFrom)) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'From': does not match with P-Preferred-Identity");

          }

        }

      }

      case (A_2_1_A5) {

        if (v_MessageHeader.callId.callid != p_InviteRequest.msgHeader.callId.callid) {

          FatalError(__FILE__, __LINE__, "A.2.1 condition A5: incorrect Call-ID");

        }

        if (v_MessageHeader.cSeq.seqNumber != vc_IMS_Global.ImsInfo.lastSeenValueOfCseq + 1) {

            FatalError(__FILE__, __LINE__, "A.2.1 condition A5: incorrect Cseq");

        }

        if (p_IMS_INVITE_REQ.Request.Invite.msgHeader.fromField != p_InviteRequest.msgHeader.fromField) { //this is strict as we request exactly the same From header as it appeared in original INVITE

            FatalError(__FILE__, __LINE__, "A.2.1 condition A5: incorrect From header");

        }

        if (p_IMS_INVITE_REQ.Request.Invite.requestLine.requestUri != p_183_header.contact.contactBody.contactAddresses[0].addressField.nameAddr.addrSpec) {

            FatalError(__FILE__, __LINE__, "A.2.1 condition A5: incorrect Request-URI");

        }        

      }

      case (A_2_1_A6) {

        v_IMEI := hex2str(px_IMEI_Def); // @sic R5s130756 change 11 sic@

         v_SipInstanceValue := """" & "<urn:gsma:imei:" & substr(v_IMEI, 0, 8) & "-" & substr(v_IMEI, 8, 6) & "-" & substr(v_IMEI, 14, 1) & ">" & """";

        if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("+sip.instance", v_SipInstanceValue))) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "IMEI code of the UE, coded as specified in draft-montemurro-gsma-imei-urn, not matching");

        }

      }

      case (A_2_1_A7) {

        v_ExpectedFromUrl := f_IMS_PTC_ImsInfo_GetSipUriRX(emergencyId, *);

        if (not match(v_SipUrlFrom, v_ExpectedFromUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'From'");

        }

        v_SipUrlPreferred := f_Addr_Union_GetSipUrl(v_MessageHeader.pPreferredID.pPreferredIDValueList[0]);

        if (not match(v_SipUrlPreferred, v_ExpectedFromUrl)) {

          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'P-Preferred-Identity'");

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    // ****  Check further Contact feature-param acc. to further conditions of A.2.1:  ****

    //

    if (p_A3) {                         // A.1.2 condition A3: MTSI

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted)) or        // @sic R5s130738: string needs to be quoted sic@

          not match(v_AcceptContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue_Quoted))) {  // @sic R5s130738: string needs to be quoted sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param");

      }

    }

    if (p_A8) {                         // A.1.2 condition A8: check geolocation and geolocationRouting @sic R5-133151 sic@

      /* NOTE: v_MessageHeader.geolocation and geolocationRouting are checked already by template returned by f_IMS_InviteRequest_MessageHeaderRX:

       *       geolocation and geolocationRouting to be present. @sic R5-133151 sic@ */

//      FatalError(__FILE__, __LINE__, "A.2.1 condition A8: geolocation needs to be checked regarding correct setting of locationURI (not implemented yet)");

      if (not match(v_MessageHeader.geolocationRouting.state, GEOLOCATION_ROUTING_YES_E)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A8: Wrong geolocation-routing");

      }

    }

    if (p_A10) {                        // A.1.2 condition A10: UE indicates video media feature tag in REGISTER and INVITE request

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10: Missing feature-param");

      }

    }

    if (p_A11) {                        // A.1.2 condition A11: INVITE for creating a video call and UE supports video media feature tag

      if (not match(v_AcceptContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A11: Missing feature-param");

      }

    }

    if (p_A12) {                        // A.1.2 condition A12: INVITE for creating a voice call and UE supports g.3gpp.srvcc-alerting media feature tag

      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("g.3gpp.srvcc-alerting", omit))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A12: Missing feature-param");

      }

    }

  }
Change 5 – f_IMS_MOCallSetup_AnnexC21C25_Step9_11
	Function name
	f_IMS_MOCallSetup_AnnexC21C25_Step9_11

	Reason for change
	Function f_IMS_MOCallSetup_AnnexC21C25_Step9_11 is called before the actual 15.11 flow starts. In this function the PRACK request is received and checked. Note that PRACK is the last request incrementing the Cseq counter before starting the actual 15.11 flow. 

	Summary of change
	Extracted the Cseq counter from the PRACK request and stored it in the global variable defined in Change 1. 
Note: in this CR, we introduced the feature of having universal access to the Cseq counter. If a correct value of this counter should be available at all times for all test cases, we would need to check the handling of all in-dialog requests and see if we would need to update the counter. This CR does this only for the task at hand, i.e., for TC 15.11.

	TTCN module
	Common/IMS/IMS_Procedures_CallControl

	MCC160 Comment
	


Before change

  function f_IMS_MOCallSetup_AnnexC21C25_Step9_11(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                                  charstring p_ContactUri := px_IMS_CalleeContactUri) runs on IMS_PTC

  { /* @sic R5s130333 change 9.2 - parameter p_ContactUri: acc. to see A.2.4 contact in request line of PRACK shall be the same in as in previous response (180 Ringing) sic@ */

    var INVITE_Request v_InviteRequest := p_IMS_INVITE_REQ.Request.Invite;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_IMS_INVITE_REQ.RoutingInfo);

    var template (value) MessageHeader v_MessageHeader_180Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, p_ContactUri);

    var template (value) MessageHeader v_MessageHeader_PrackResponse_200;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var PRACK_Request v_PrackRequest;

    // Step 9. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_180Response)));

    // Step 10. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_180Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeader_Prack))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    //fl_IMS_Params_CheckSigComp(v_PrackRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 9 (PRACK) "comp=sigcomp" is optional for Via and Route

    // Step 11. Send 200 OK (response to last PRACK)

    v_MessageHeader_PrackResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_PrackResponse_200)));

  }
After change

  function f_IMS_MOCallSetup_AnnexC21C25_Step9_11(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                                  charstring p_ContactUri := px_IMS_CalleeContactUri) runs on IMS_PTC

  { /* @sic R5s130333 change 9.2 - parameter p_ContactUri: acc. to see A.2.4 contact in request line of PRACK shall be the same in as in previous response (180 Ringing) sic@ */

    var INVITE_Request v_InviteRequest := p_IMS_INVITE_REQ.Request.Invite;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_IMS_INVITE_REQ.RoutingInfo);

    var template (value) MessageHeader v_MessageHeader_180Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, p_ContactUri);

    var template (value) MessageHeader v_MessageHeader_PrackResponse_200;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var PRACK_Request v_PrackRequest;

    // Step 9. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_180Response)));

    // Step 10. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_180Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeader_Prack))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    vc_IMS_Global.ImsInfo.lastSeenValueOfCseq := v_PrackRequest.msgHeader.cSeq.seqNumber; 

    //fl_IMS_Params_CheckSigComp(v_PrackRequest.msgHeader.route.routeBody[0].rrParam);     // acc. to 34.229-1 cl. 13.2.4 step 9 (PRACK) "comp=sigcomp" is optional for Via and Route

    // Step 11. Send 200 OK (response to last PRACK)

    v_MessageHeader_PrackResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_PrackResponse_200)));

  }
5. Execution Log Files

Intel XMM7160 UE passed this test case on the Rohde & Schwarz Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_15_11_Intel_XMM7160.log

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6. References

[1] R5s140018: Supporting information for re-verification of IMS test case 15.11 over LTE with 34.229-3 test model - This archive comprises txt format execution log file
