
Change 1 
	Function name
	Function f_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	According to 24.229 5.1.6.8.2, a UE establishing an emergency session without prior registration should include use the display name “anonymous” in the “from” hader of the INVITE request, as specified in RFC 3261, which gives the example : 

From: Anonymous <sip:c8oqz84zk7z@privacy.org>;tag=hyh8

However 24.607, and also other RFCs (i.e. RFC 3323) give an example in the format  (with quotes):  

From: "Anonymous" <sip:anonymous@anonymous.invalid>;tag= xxxxxxx.

From the SIP syntax in RFC 3621,the Display-Name can be either a set of tokens or a quoted string.

display-name   =  *(token LWS)/ quoted-string

So the TTCN should accept the display name “anonymous” either with or without quotes.

	Summary of change
	Created v_Displayname1 and v_Displayname2 to store value Anonymous, with and without quotes and this is passed as a list in cr_From template to handle this.

	TTCN module
	IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before change

    function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI

       p_A8:          UE uses Geolocation header to provide its geographical location for emergency session setup, has obtained its location and is setting up an emergency session

       p_A9:          UE supports Session-ID

       p_A11:         INVITE for creating a video call and UE supports video media feature tag

       NOTE1: If SigComp shall be applied the param "comp=sigcomp" is optional for Via, Route and Contact

       NOTE2: condition A5 (re-INVITE within a dialog) is not considered yet and will have impact e.g. on checking of the From/To headers */

    /* @sic R5-133629: 100rel option tag shall always be included in INVITE; further option tags may be added => p_Supported replaced by p_AdditionalOptionTagsForSupported; Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    var template integer v_Port_ps := *;

    var template charstring v_Displayname := *;
    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_MediaType := "application/sdp";

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_Displayname := "Anonymous";
    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    // route body acc. to conditions in A.2.1

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

    if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

      v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

    }

    if (p_Context == A_2_1_A6) { // Emergency call limited services 

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Pcscf, v_Port_ps)); // #ob#BZ49422

    }

//R5s130917

    if (p_A8) {

      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

      v_MediaType := "multipart/mixed";

    }

    v_MessageHeader_Invite.via             := fl_IMS_InviteRequest_ViaRX(v_IsGiba);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := cr_From(v_Displayname, v_SipUrlFrom);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri);                          /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := cr_CallId(?);                                           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := f_IMS_SessionId_RX_IfSupported(p_A9);                   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(?, "INVITE");                                /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or (p_Context == A_2_1_A6)) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (p_Context == A_2_1_A6) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Invite.require         := cr_RequireSecAgree;

      v_MessageHeader_Invite.proxyRequire    := cr_ProxyRequireDef;

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    }

    if (p_Context == A_2_1_A6) {

      v_MessageHeader_Invite.pAccessNetworkInfo := *; // #ob#BZ49422 FFS

    }    

//R5s130917

    if (p_Context != A_2_1_A4) {

      v_MessageHeader_Invite.accept          := cr_Accept({cr_AcceptBody("application/sdp"), cr_AcceptBody("application/3gpp-ims+xml"), *});

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3) {

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_AcceptContactValue);

    }

    if (p_A3 or p_A11) {

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  } 
After change

  function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI

       p_A8:          UE uses Geolocation header to provide its geographical location for emergency session setup, has obtained its location and is setting up an emergency session

       p_A9:          UE supports Session-ID

       p_A11:         INVITE for creating a video call and UE supports video media feature tag

       NOTE1: If SigComp shall be applied the param "comp=sigcomp" is optional for Via, Route and Contact

       NOTE2: condition A5 (re-INVITE within a dialog) is not considered yet and will have impact e.g. on checking of the From/To headers */

    /* @sic R5-133629: 100rel option tag shall always be included in INVITE; further option tags may be added => p_Supported replaced by p_AdditionalOptionTagsForSupported; Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    var template integer v_Port_ps := *;

    //var template charstring v_Displayname := *;

    var template charstring v_Displayname1 := *;

    var template charstring v_Displayname2 := *;
    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_MediaType := "application/sdp";

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_Displayname1 := """Anonymous""";

      v_Displayname2 := "Anonymous";
    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    // route body acc. to conditions in A.2.1

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

    if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

      v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

    }

    if (p_Context == A_2_1_A6) { // Emergency call limited services 

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Pcscf, v_Port_ps)); // #ob#BZ49422

    }

//R5s130917

    if (p_A8) {

      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

      v_MediaType := "multipart/mixed";

    }

    v_MessageHeader_Invite.via             := fl_IMS_InviteRequest_ViaRX(v_IsGiba);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := cr_From((v_Displayname1, v_Displayname2), v_SipUrlFrom);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri);                          /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := cr_CallId(?);                                           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := f_IMS_SessionId_RX_IfSupported(p_A9);                   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(?, "INVITE");                                /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or (p_Context == A_2_1_A6)) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (p_Context == A_2_1_A6) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Invite.require         := cr_RequireSecAgree;

      v_MessageHeader_Invite.proxyRequire    := cr_ProxyRequireDef;

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    }

    if (p_Context == A_2_1_A6) {

      v_MessageHeader_Invite.pAccessNetworkInfo := *; // #ob#BZ49422 FFS

    }    

//R5s130917

    if (p_Context != A_2_1_A4) {

      v_MessageHeader_Invite.accept          := cr_Accept({cr_AcceptBody("application/sdp"), cr_AcceptBody("application/3gpp-ims+xml"), *});

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3) {

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_AcceptContactValue);

    }

    if (p_A3 or p_A11) {

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  } 

