
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 16.1 which is part of the ‘IMS_34229_IWD_13wk35’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk35’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
16.1
Test Group:
IMS Codec selecting
ATS Version:
iwd-EUTRA-B2012-03_D13wk35
System Simulator used:
Rohde & Schwarz Multi-RAT Protocol Tester
UE used:
Intel XMM7160
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk35’ release.
Change 1 – f_TC_16_1_and_16_2_TestBody
	Function name
	f_TC_16_1_and_16_2_TestBody

	Reason for change
	1. f_IMS_CC_StartCall already called in embedding function


2. Test case is modeled according to flow in section 5.6.3 of 24.940. This means that UE has resources at start of flow
3. Call to accepting the call comes too early

	Summary of change
	1. Removed call
2. Moved call to set up dedicated bearer from bottom of the flow to the top
3. Moved call down under step 6 to trigger UE to send the 200OK for INVITE then

	TTCN module
	IMS/16/IMS_CC_CodecSelecting

	MCC160 Comment
	


Before change

function f_TC_16_1_and_16_2_TestBody(boolean p_IsGIBA,

                                       template (value)   SDP_Message p_SDP_Message_Step1,

                                       template (present) SDP_Message p_SDP_Message_Step4or7) runs on IMS_PTC

  { // !!!! timer needs to be removed !!!!!!!!!

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var Response v_Response;

    var MessageHeader v_MessageHeader;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (present) MessageHeader v_MessageHeader_180RX;

    var template SDP_Message v_SDP_Message_Step7 := omit;

    var SDP_Message v_SDP_Message_Step4or7;

    var boolean v_Trying := false;

    timer t_Timer := 5.0;   // as for C.11 (-> f_IMS_MTCallSetup_Common_Steps7_11)

    f_IMS_CC_StartCall(IPCAN_MT_SpeechCall);
    // Step 1 - Send INVITE

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX(tsc_OptionTagList_precondition, p_IsGIBA);

    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(p_SDP_Message_Step1));

    IMS_Client.send(cas_IMS_Invite_Request(cs_IMS_RoutingInfo(v_Protocol), v_InviteRequest));

    // Step 2 - Make UE accept the speech AMR offer

    f_UT_IMS_AcceptCall(MMI);        // NOTE: the MMI command only triggers the UE to accept the call but we don't know at which point in time the UE gets the command

    v_MessageHeader_180RX := f_IMS_InviteResponse_180_MessageHeaderRX(v_InviteRequest, p_IsGIBA);

    v_MessageHeader_180RX.contentType := cr_ContentTypeDef("application/sdp") ifpresent;

    t_Timer.start;

    alt {

      // Step 3 - Receive optional 100 Trying

      [not v_Trying] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderRX(v_InviteRequest))))

        {

          v_Trying := true;

          repeat;

        }

      // Step 4 - Receive optional 180 Ringing

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180,

                                                                                         v_MessageHeader_180RX,

                                                                                         cr_MessageBody_SDP(p_SDP_Message_Step4or7) ifpresent)))

        -> value v_IMS_DATA_RSP

        {

          t_Timer.stop;

          v_Response := v_IMS_DATA_RSP.Response;

          v_MessageHeader := v_Response.msgHeader;

          f_IMS_Dialog_SetToTag(v_MessageHeader);

          if (ischosen(v_Response.messageBody.sdpMessageBody)) {

            v_SDP_Message_Step4or7 := v_Response.messageBody.sdpMessageBody;

            if (not ispresent(v_MessageHeader.contentType)) {

              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: missing content type");

            }

          }

        }

      [] t_Timer.timeout {}

    }

    if (isbound(v_MessageHeader)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader, c_tag100rel)) {

        // Step 5 - 6. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader);

      }

    }

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer
    if (not isbound(v_SDP_Message_Step4or7)) {

      v_SDP_Message_Step7 := p_SDP_Message_Step4or7;

    }

    // Step 7 - 8. Receive OK for INVITE, Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.

    v_Response := f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequest, v_Protocol, v_SDP_Message_Step7, p_IsGIBA);

    v_MessageHeader := v_Response.msgHeader;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_MessageHeader);

    if (not isbound(v_SDP_Message_Step4or7)) {   // i.e. there has been no 180 Ringing or the 180 has not included an SDP message

      v_SDP_Message_Step4or7 := v_Response.messageBody.sdpMessageBody;

    }

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message_Step4or7);

    // Step 9-10

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl);

    f_IMS_CC_ReleaseConnection(IPCAN_MT_SpeechCall);

  }
After change

   function f_TC_16_1_and_16_2_TestBody(boolean p_IsGIBA,

                                       template (value)   SDP_Message p_SDP_Message_Step1,

                                       template (present) SDP_Message p_SDP_Message_Step4or7) runs on IMS_PTC

  { // !!!! timer needs to be removed !!!!!!!!!

    var SipUrl v_ContactUrl := f_IMS_PTC_ImsInfo_GetContactUrl();

    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var Response v_Response;

    var MessageHeader v_MessageHeader;

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (value) INVITE_Request v_InviteRequest;

    var template (present) MessageHeader v_MessageHeader_180RX;

    var template SDP_Message v_SDP_Message_Step7 := omit;

    var SDP_Message v_SDP_Message_Step4or7;

    var boolean v_Trying := false;

    timer t_Timer := 5.0;   // as for C.11 (-> f_IMS_MTCallSetup_Common_Steps7_11)

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MT_SpeechCall); 
    // Step 1 - Send INVITE

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderTX(tsc_OptionTagList_precondition, p_IsGIBA); //HRTODO: add support for 100rel according to 34.229-1, A.2.9

    v_InviteRequest := cs_INVITE_Request(v_ContactUrl, v_MessageHeader_Invite, cs_MessageBody_SDP(p_SDP_Message_Step1));

    IMS_Client.send(cas_IMS_Invite_Request(cs_IMS_RoutingInfo(v_Protocol), v_InviteRequest));

    // Step 2 - Make UE accept the speech AMR offer

    v_MessageHeader_180RX := f_IMS_InviteResponse_180_MessageHeaderRX(v_InviteRequest, p_IsGIBA);

    v_MessageHeader_180RX.contentType := cr_ContentTypeDef("application/sdp") ifpresent;

    t_Timer.start;

    alt {

      // Step 3 - Receive optional 100 Trying

      [not v_Trying] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderRX(v_InviteRequest))))

        {

          v_Trying := true;

          repeat;

        }

      // Step 4 - Receive optional 180 Ringing

      [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_Protocol), cr_Response(c_statusLine180,

                                                                                         v_MessageHeader_180RX,

                                                                                         cr_MessageBody_SDP(p_SDP_Message_Step4or7) ifpresent)))

        -> value v_IMS_DATA_RSP

        {

          t_Timer.stop;

          v_Response := v_IMS_DATA_RSP.Response;

          v_MessageHeader := v_Response.msgHeader;

          f_IMS_Dialog_SetToTag(v_MessageHeader);

          if (ischosen(v_Response.messageBody.sdpMessageBody)) {

            v_SDP_Message_Step4or7 := v_Response.messageBody.sdpMessageBody;

            if (not ispresent(v_MessageHeader.contentType)) {

              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: missing content type");

            }

          }

        }

      [] t_Timer.timeout {}

    }

    if (isbound(v_MessageHeader)) {

      f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader);      // check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)

      if (f_MessageHeader_CheckRequire(v_MessageHeader, c_tag100rel)) {

        // Step 5 - 6. Send PRACK, Receive 200 OK if the 180 response contains 100rel option tag within the Require header

        f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_MessageHeader);

      }

    }

    if (not isbound(v_SDP_Message_Step4or7)) {

      v_SDP_Message_Step7 := p_SDP_Message_Step4or7;

    }

    f_UT_IMS_AcceptCall(MMI); 
    // Step 7 - 8. Receive OK for INVITE, Send ACK. The SS acknowledges the receipt of 200 OK for INVITE.

    v_Response := f_IMS_MTCallSetup_Common_Steps12_13(v_InviteRequest, v_Protocol, v_SDP_Message_Step7, p_IsGIBA);

    v_MessageHeader := v_Response.msgHeader;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_MessageHeader);

    if (not isbound(v_SDP_Message_Step4or7)) {   // i.e. there has been no 180 Ringing or the 180 has not included an SDP message

      v_SDP_Message_Step4or7 := v_Response.messageBody.sdpMessageBody;

    }

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message_Step4or7);

    // Step 9-10

    f_IMS_CallReleaseMT(v_InviteRequest, v_ContactUrl);

    f_IMS_CC_ReleaseConnection(IPCAN_MT_SpeechCall); 

  }
Change 2 – f_IMS_BuildSDP_AMR_RX
	Function name
	f_IMS_BuildSDP_AMR_RX

	Reason for change
	TC calls this function omitting the second argument. As this argument feeds the check for what will be accepted on the a=fmtp line the default value for this argument should not be empty (as otherwise only the media type number would be allowed)

	Summary of change
	Allow wildcard as default argument

	TTCN module
	IMS/16/IMS_CC_CodecSelecting

	MCC160 Comment
	


Before change

  function f_IMS_BuildSDP_AMR_RX(charstring p_Fmt,

                                 CharStringList_Type p_MediaFormatParameters := {}) runs on IMS_PTC

    return template (present) SDP_Message

  {

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_Fmt, tsc_CODEC_AMR_8000_1),

      cr_SDP_Attribute_fmtp(p_Fmt, p_MediaFormatParameters),

      // precondition attributes:

      cr_SDP_Attribute_curr_quos(c_local,  c_sendrecv),

      cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv),

      cr_SDP_Attribute_des_quos(c_mandatory, c_local),

      cr_SDP_Attribute_des_quos(c_mandatory, c_remote), // @sic R5s120530 change 3.4 sic@

      *

    };

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_Fmt),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
After change

  function f_IMS_BuildSDP_AMR_RX(charstring p_Fmt,

                                 CharStringList_Type p_MediaFormatParameters := {"*"}) runs on IMS_PTC  

    return template (present) SDP_Message

  {

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template (present) SDP_attribute_list v_SDP_Attribute_List := {

      // media attributes:

      cr_SDP_Attribute_rtpmap(p_Fmt, tsc_CODEC_AMR_8000_rx), 

      cr_SDP_Attribute_fmtp(p_Fmt, p_MediaFormatParameters),

      // precondition attributes:

      cr_SDP_Attribute_curr_quos(c_local,  c_sendrecv),

      cr_SDP_Attribute_curr_quos(c_remote, c_sendrecv),

      cr_SDP_Attribute_des_quos(c_mandatory, c_local),

      cr_SDP_Attribute_des_quos(c_mandatory, c_remote), // @sic R5s120530 change 3.4 sic@

      *

    };

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(p_Fmt),

                                          cr_SDP_Connection(v_AddrType, v_IPAddrConnection),

                                          cr_SDP_Bandwidth_List_Media(?, ?, ?),

                                          v_SDP_Attribute_List);

    return cr_SDP_Message_Def(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }
5. Execution Log Files

Intel XMM7160 UE passed this test case on the Rohde & Schwarz Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_16_1_XMM7160.log

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6. References

[1] R5s130809: Supporting information for re-verification of IMS test case 16.1 over LTE with 34.229-3 test model - This archive comprises html format execution log file
